


## **General Disclaimer**

### **One or more of the Following Statements may affect this Document**

- This document has been reproduced from the best copy furnished by the organizational source. It is being released in the interest of making available as much information as possible.
- This document may contain data, which exceeds the sheet parameters. It was furnished in this condition by the organizational source and is the best copy available.
- This document may contain tone-on-tone or color graphs, charts and/or pictures, which have been reproduced in black and white.
- This document is paginated as submitted by the original source.
- Portions of this document are not fully legible due to the historical nature of some of the material. However, it is the best reproduction available from the original submission.



DEPARTMENT OF PHYSICS AND GEOPHYSICAL SCIENCES  
SCHOOL OF SCIENCES AND HEALTH PROFESSIONS  
OLD DOMINION UNIVERSITY  
NORFOLK, VIRGINIA

Technical Report PGSTR-AP75-15

(NASA-CF-147986) SOFTWARE FOR DIGITAL  
ACQUISITION SYSTEM AND APPLICATION TO  
ENVIRONMENTAL MONITORING (Old Dominion  
Univ., Norfolk, Va.) 47 p HC \$4.00 CSCL 13B

N76-24750

Unclas  
G3/45 28189


SOFTWARE FOR DIGITAL ACQUISITION SYSTEM AND  
APPLICATION TO ENVIRONMENTAL MONITORING

By

G.E. Copeland

Funded by  
NASA Grant NGL 47-003-067

November 1975



DEPARTMENT OF PHYSICS AND GEOPHYSICAL SCIENCES  
SCHOOL OF SCIENCES AND HEALTH PROFESSIONS  
OLD DOMINION UNIVERSITY  
NORFOLK, VIRGINIA

Technical Report PGSTR-AP75-15

SOFTWARE FOR DIGITAL ACQUISITION SYSTEM AND  
APPLICATION TO ENVIRONMENTAL MONITORING

*By*

G.E. Copeland

*Funded by*

NASA Grant NGL 47-003-067

Principal Investigator: Earl C. Kindle  
Professor of Physics and Geophysical Sciences

Technical Monitor: J.A. Vitale  
NASA Office of University Affairs  
Washington, DC 20546

*Submitted by the*  
Old Dominion University Research Foundation  
Norfolk, Virginia 23508



November 1975

## TABLE OF CONTENTS

	<u>Page</u>
ABSTRACT . . . . .	1
INTRODUCTION . . . . .	2
HARDWARE CONFIGURATION . . . . .	3
DATA ACQUISITION SOFTWARE . . . . .	4
FUTURE EXPANSION PLANS . . . . .	7
TABLE . . . . .	10
FIGURES . . . . .	11
REFERENCES . . . . .	21
APPENDIX: MONARCH WRITTEN IN PAL III	

## ABSTRACT

Criteria for selection of a mini-computer for use as a core resident acquisition system are developed for the ODU Mobile Air Pollution Laboratory. A comprehensive data acquisition program named MONARCH has been instituted in a DEC-8/E-8K 12-bit computer. Up to 32 analog voltage inputs are scanned sequentially, converted to BCD, and then to actual numbers. As many as 16 external devices (valves or any other two-state device) are controlled independently. MONARCH is written as a foreground-background program, controlled by an external clock which interrupts once per minute. Transducer voltages are averaged over user specified time intervals and, upon completion of any desired time sequence, outputted are: day, hour, minute, second; state of external valves; average value of each analogue voltage (E Format); as well as standard deviations of these values. Output is compatible with any serially addressed media.

## INTRODUCTION

The need for development of a computer based data acquisition and control system arose during the evolution of a comprehensive Mobile Air Pollution Laboratory at Old Dominion University. This laboratory, housed in a mobile trailer (see figure 1), monitors at four different heights the following environmental parameters: wind speed and direction; temperature; concentrations of ozone ( $O_3$ ); nitrogen oxides ( $NO$ ,  $NO_2$ ,  $NO_x$ ); carbon monoxide ( $CO$ ); total hydrocarbons ( $THC$ ); and sulfur compounds ( $H_2S$ ,  $SO_2$ ). Additionally measurements at one height are: relative humidity; insolation; atmospheric pressure; and the b-scattering coefficient (visibility). All of these analog voltage instruments were initially measured sequentially by a multipoint recorder which produced 14,000 data points per level per day. Transcription and averaging of this data base required more effort than maintenance of the instruments during the course of month-long field experiments. It soon became obvious that a computer based data acquisition and control system must be installed to provide near real time analysis of the environmental parameters and to provide convenient expansion and modification via software when additional instrumentation became available. This system was named MONARCH.

A detailed analysis of present and future requirements for MONARCH suggested the following hardware configuration.

1. Data acquisition. A mixture of analog voltage, bridges, and digital BCD input channels must be available. The software must select order and frequency of scanning for a high precision low-speed A/D. All inputs to the A/D are set in the range -10.00 to +10.00 volts, so that each input slot has its own amplifier to change the incoming voltages to this range. Cost constraints prohibited use of a programmable gain amplifier.

2. All data averaging is done in software.

3. Master Control. A crystal controlled clock provides interrupts once per minute. Software design is based upon counting clock interrupts. When read, the clock provides Julian date, hour, minute, and second in three 12-bit BCD words.

4. Control Functions. Computer must initiate all scans of instruments, read the clock, and be able to control external valves for level shifting as well as inserting scrubbers, span and zero gases at appropriate times into air pollution instruments.

5. Input/Output. The entire system should work in a conversational mode so that field personnel may control its actions. Output medium should be versatile so that changes from paper tape to cassette tape or to tele-communications is possible.

Criteria were established for selection of a computer for MONARCH. Realizing that none of the investigators had ever dealt with minicomputers or assembly languages, one of the major criteria for selection was the existence of clearly worded comprehensible software documentation. Other criteria were: ease of interfacing, possibility of future expansion, cost, availability of service and software personnel, and our in-house hardware capability. The final selection of the minicomputer was a PDP8/E with 8K 12-bit word core memory.

#### HARDWARE CONFIGURATION

A block diagram of the hardware system is shown in figure 2. All I/O is handled by a 33ASR teletype. The clock was built in-house and interfaced to the PDP8/E. The clock (device code 13) interrupts once per minute and provides day (000-365), hour (00-24), minute and second (00-60) in three 12-bit words. It can be reset manually to 000<sup>d</sup>10<sup>h</sup>00<sup>m</sup>00<sup>s</sup>.

The valve control unit (device code 15) accepts from the PDP8/E one 12-bit word. Each bit (0 or 1) controls the state of one valve system. Once a valve is set via software command it maintains that state until it is reset. This device is expandable to 36 valve states.

Since MONARCH can control measurements at four levels (15, 25, 50, 75 feet) three of the valve states are used (figure 3). Four glass and teflon manifolds are brought into the trailer where each molecular instrument is attached to each manifold via 1/4" I.D. teflon tubing. Switching is accomplished by valves A, B, C which are solenoid activated teflon 3-way divert valves. This arrangement leaves eight unused states for cycling of scrubbers. If zero air or span gases are desired, they are easily incorporated with these additional valve states.

Device 14 is the A/D scanner and input level shifters for 14 input data channels. It is expandable up to 18 additional analog channels. Channel selection is via transfer of a positive binary number from the accumulator to the decoder where the appropriate channel is selected and its mercury relay is closed and data conversion begins. Upon finish of data conversion this device interrupts and one 12-bit BCD number can be read by the computer. Thus this device both accepts and sends information. If digital information is available from any instrument, they can be assigned their own device codes as needed.

#### DATA ACQUISITION SOFTWARE

MONARCH operates with the interrupt enabled, and a background program which endlessly rotates one bit through the link and accumulator of the PDP8/E. When the indicator selector switch is in the AC setting, one light moves across the display at a rate and direction dependent upon the switch registers (1). Any similar program can be substituted.



An initial start up sequence, START, operates with interrupt off and a dialogue takes place between the operator and MONARCH (figure 4). This sequence sets the number of levels, instruments, samples per instrument per minute, the length of time MONARCH spends per level and time periods between zero and span cycles. At the end of START all software counters are initialized. MONARCH gives messages to the operator, sets the valves and then halts.

MONARCH operates using a Digital Equipment Corporation 23-bit floating point package (FPP) (2), series of teletype service routines, and a BCD-binary conversion subroutine. Use of the FPP enables accuracy of six significant digits and greatly enhances the ease of arithmetic and data manipulation as well as I/O operations. Figures 5 through 10 give the flow diagram for MONARCH. A listing of MONARCH in PAL III assembly language constitutes Appendix A.

After the operator has interacted with MONARCH and supplied the required information, the computer halts. Upon pressing the CONTINUE switch, all flags are cleared, the interrupt is turned on and the background program is entered. At this point we will assume four levels, 14 instruments, five minutes per level and 100 samples per instrument per minute. At the first clock interrupt (Zeroth minute) MONARCH goes to location 0000 and executes the instruction stored in location 0001 which is an effective jump to the Service routine. The Service routine stores the current value of the accumulator and link. Location 0000 has stored in it the next instruction that was to be performed before interrupt occurred. The SERVE routine is a skip chain which tests to see which device needs attention. At this moment the clock is determined and a jump to CLKSER occurs. Since this is the first clock interrupt, the clock is not read. Several counters are incremented (counted up to zero), and the scanner is instructed to start conversion on instrument 14. Then the program jumps to EXIT where the accumulator and link are restored, and program control is returned to background.

The next interrupt is the scanner signalling that instrument 14 has finished conversion. A jump to SERVE occurs where the scanner is detected and a jump to SCNSER happens. SCNSER clears the scanner interrupt flag, sends one 12-bit BCD word to the accumulator. Since the A/D has been designed to accept -10 to +10 volt signals, it is necessary to determine the algebraic sign of the data. This is implemented in hardware. If the least significant digit (LSD) is 0, then the number is negative. If 1, it is positive. This means valid outputs from the A/D are even numbers (negative); odd numbers (positive) over the range -998 to +999. Over range is signalled by all bits zero and must be differentiated from real zero by software if needed.

SCNSER checks the LSD and converts BCD to binary via a service routine, BCDBIN. The floating point package (FPP) is entered and the one word binary data is converted to floating point (3 words). This data is level shifted to all positive (range 0 to 2000) and added indirectly to the 14th floating point buffer location. The Buffer was set to zero in START. Next the scanner is sent instructions to process instrument 13 and control is passed to background.

This process continues until SCNSER detects the fact that each instrument has been read the specified number of times (100, in this example). At that point the scanner is not sent a start conversion signal and program returns to background to wait for the second clock interrupt.

This procedure (clock interrupt, read sequentially each instrument 100 times, wait until clock interrupts) continues until the fourth clock interrupt. At that time a valve in the NO<sub>x</sub> instrument is energized and the same procedure continues.

At the fifth clock interrupt a new sequence is initialed. First, the valve state (level) is changed. The clock is now read (BCD 3 words). The clock words are masked and converted to binary, and then to 12-bit words corresponding to units of seconds, minutes, days, tens of seconds, minutes, days, and hundreds

of days. Next the state of the NO<sub>x</sub> valve is changed. Data output is now indicated via the routine DTAOUT. An example output is shown in table 1. Extensive use is made of the FPP and teletype routines for output and format control (may be FORTRAN E or F format). Output is sequentially printed on three TTY lines; Line 1: DAY, HOUR, MIN., SEC., VALVE LEVEL CODE. Line 2: DATA FROM INSTRUMENT ONE TO SEVEN. Line 3: DATA FROM INSTRUMENT EIGHT TO FOURTEEN. Each outputted value is a decimal number and each are separated by blanks except carriage return line feed at the end of lines. The values are the arithmetic mean of the voltages of each instrument averaged over the number of times it was read. DTAOUT then zeroes all the data storage buffer and returns to background via EXIT. Since the output is to TTY paper tape, a maximum of 600 characters may be punched before the next interrupt occurs. DTAOUT operates with the interrupt off, so that no confusion is possible.

Included in the SERVE Skip Chain are two other service routines. TTYSER gives a warning to the operator if the TTY keyboard is struck. This can be used to input and modify future states of the program. ERROR is a routine at the bottom of the skip chain which can only be entered if interrupt occurs when no real device has caused such. This routine gives an error message and returns to background. It is of importance in cases where the electromagnetic environment is noisy.

#### FUTURE EXPANSION PLANS

Even though MONARCH presently resides in an 8K machine, it is entirely contained in Field 0 (Lower 4K memory). The upper 4K memory is not used. Thus, MONARCH can be used with only minor modification in any 4K PDP 8 system. Since the instruction set utilized is shared by the PDP/5 system, MONARCH will work in these older machines if the interrupt service routine is modified. If mass

storage devices are incorporated for I/O operations, then the TTY service routines and FPP FOUT function must be modified.

The present version of MONARCH is fixed to a previously existing four-level value structure shown in figure 3. If the number of external levels are increased, additional value routines are necessary. Decreasing from four levels requires no change. If zero and span cycles are desired, they may be inserted in the ZERO routine or they may be assigned levels five and six and thus will automatically be cycled through in the same time intervals as the first four levels.

Many of the instructions used in the floating point package are not currently used. Additional storage may be obtained by deleting these. Examples would be the floating trig and log functions. One modification that would be desirable, from a data analysis viewpoint, would be the incorporation of statistical variances of the quantities being measured. This could be implemented without greatly increasing data storage requirements (it would double -  $14 \times 3 \times 2 = 84 - 12$  bit words). Presently, the data storage buffer has stored in each location (3 word) the sum of the data from each instrument over the sample interval. Using the FSQU instruction and doubling the size of the buffer, one could have stored

$$\sum_{i=1}^N x_i$$

and

$$\sum_{i=1}^N x_i^2$$

where  $x_i$  is the measurement,  $N$  is the number of times the quantity is measured. At output, it is a simple matter to output (already done)

$$\langle x \rangle = \frac{1}{N} \sum_{i=1}^N x_i$$

and also

$$\sigma^2 = \langle x^2 \rangle - \langle x \rangle^2 = \frac{1}{N} \sum x_i^2 - \left( \frac{1}{N} (\sum x_i) \right)^2$$

The variances have physical interpretations for meteorological quantities (wind speed, temperature, etc.) which are related to turbulence. Variances should be known so that estimates of the reliability and validity of the data may be obtained.

If reliable calibration equations are available, the floating point package can be used to provide output in engineering units directly. Although this may appear to be a desired goal in design of MONARCH, it can easily produce invalid results if calibration of any instrument changes. As a check against this happening for the computer hardware, it is suggested that at least one data port be used to monitor a fixed voltage (Standard cell).

Table 1. Example of MONARCH output.

Line 1	Day	Hour	Min	Sec	Valve State		
Line 2	Inst # 1	2	3	4	5	6	7
Line 3	8	9	10	11	12	13	14

+401.0	+14.0	+53.0	+0.0	+5.			
+75.4	+75.3	+75.3	+75.2	+75.3	+75.3	+75.3	
+75.3	+75.3	+58.9	+69.7	+64.7	+67.0	+1874.9	

+401.0	+14.0	+58.0	+0.0	+4.			
+76.9	+78.6	+77.2	+78.8	+77.3	+78.8	+77.3	
+78.8	+77.3	+62.1	+73.2	+69.0	+70.2	+1872.8	

+401.0	+15.0	+3.0	+0.0	+2.			
+77.5	+77.4	+77.5	+77.6	+77.6	+77.7	+77.6	
+77.7	+77.6	+60.8	+71.9	+69.1	+69.3	+1872.9	

+401.0	+15.0	+8.0	+0.0	+0.			
+74.7	+76.0	+74.9	+76.1	+74.9	+76.2	+75.0	
+76.2	+74.9	+59.4	+69.7	+66.7	+67.1	+1871.6	

+401.0	+15.0	+13.0	+0.0	+5.			
+77.2	+78.6	+77.5	+79.0	+77.6	+79.0	+77.8	
+79.0	+77.6	+62.3	+73.0	+69.8	+70.1	+1869.1	

+401.0	+15.0	+18.0	+0.0	+4.			
+73.4	+75.9	+73.6	+76.1	+73.7	+76.1	+73.8	
+76.1	+73.7	+59.5	+69.5	+65.1	+66.6	+1868.8	

+401.0	+15.0	+23.0	+0.0	+2.			
+75.8	+79.4	+76.0	+79.5	+76.1	+79.7	+76.2	
+79.6	+76.0	+62.9	+71.0	+70.2	+68.2	+1870.2	




Figure 1. ODU Mobile Air Pollution Laboratory.




Figure 2. Block diagram of the hardware configuration of MONARCH.






Figure 3. Four level valve configuration.

I AM MONARCH  
 INPUT THE NUMBER OF DATA PORTS  
 14  
 INPUT THE NUMBER OF LEVELS  
 4  
 INPUT THE TIME(MIN) SPENT PER LEVEL  
 5  
 INPUT NUMBER OF SAMPLES TAKEN  
 BY EACH INSTRUMENT PER LEVEL  
 100  
 INPUT THE NUMBER OF TIMES UP TOWER  
 BETWEEN ZERO CYCLES  
 3  
  
 CLOCK CAN BE RESET ONLY AT  
  
 000 DAYS  
 10HR  
 00MIN  
 00SEC  
 WHEN I STOP TURN ON PUNCH AND HIT CONTINUE

Figure 4. Initial Dialogue between MONARCH  
 and OPERATOR.

ORIGINAL PAGE IS  
 OF POOR QUALITY




Figure 5. MONARCH flow diagram - start up and background.




Figure 6. MONARCH - interrupt routines.





Figure 7. CLKSER (clock service) routine.



ORIGINAL PAGE IS  
OF POOR QUALITY

Figure 8. Valve switching and output.





Figure 9. Service routines.



ORIGINAL PAGE IS  
OF POOR QUALITY

Figure 10. Service routines.



## REFERENCES

1. Introduction to Programming, Digital Equipment Corporation, Third Edition,  
May 1972, p. 3-36.
2. DEC-08-NFPPA-A-PA1.

## APPENDIX

MONARCH Written in PAL III

BUFF=4000

```
*200
ENTER,  CLA CLL
        TLS
        CLA CLL
        DCA AC
        JMS I PTYPX
        ENTMS3
        CLA CLL
        TAD CHAN
        DCA INDXR3+2
        CLL
        JMS I 7
        FGET INDXR3
        FPUT R16
ZERIST, FGET  ZERO00
        FPUT I R16
        FISZ R20 /ALL DONE?
        FJMP ZERIST /NO
        FGET INDXR3 /SET UP R16
        FPUT R16
        FEXT
        CLA CLL
        TLS
        JMS I PTYPX
        ENTMS1
        CLA CLL
        JMS I PTYPX
        ENTMS2
        HLT /STOP BEFORE ROTATE CHECK PAPER TAPE
        CAF
        ION
        JMP I PROTAT
ENTMS3, 3700 /CR-LF0
ENTMS2, 2710 /WH
0516 /EN
4011 / I
4023 / S
2417 /TO
2040 /P
2425 /TU
2216 /RN
4017 / O
1640 /N
2025 /PU
1603 /NC
1040 /H
0116 /AN
0440 /D
1011 /HI
2440 /T
```

ORIGINAL PAGE IS  
OF POOR QUALITY

0317 /CO  
1624 /NT  
1116 /IN  
2505 /UE  
3700 /-0  
ENTMS1, 3737 /--

0314 /CL  
1703 /OC  
1340 /K  
0301 /CA  
1640 /N  
0205 /BE  
4022 / R  
0523 /ES  
0524 /ET  
4017 / O  
1614 /NL  
3140 /Y  
0124 /AT  
3737 /--  
4060 / O  
6060 /OO  
4004 / D  
0131 /AY  
2337 /S-  
6160 /10  
1022 /HR  
3760 /-O  
6015 /OM  
1116 /IN  
3760 /-O  
6023 /OS  
0503 /EC  
3700 /-0  
\*110

PENTER, ENTER  
PROTAT, ROTATE  
\*2200

TTYSER, KCF  
CLA CLL  
TLS  
JMS I PTYPX  
TTYME1  
CAF  
JMP I PEXIT

TTYME1, 1305 /KE  
0520 /EP  
4031 / Y  
1725 /OU  
2240 /R  
1001 /HA  
1604 /ND  
2340 /S  
1706 /OF  
0640 /F  
2410 /TH  
0540 /E  
2424 /TT  
3140 /Y  
3700 /-0

ORIGINAL PAGE IS  
OF POOR QUALITY

```

*400
DTAOUT, CLA CLL
TLS
JMS I PTYPX /+2LF/CR
OPMSI
CLA CLL
IAC /PUT >0 IN AC
DCA 56 / TAKE OUT OF E FORMAT
TAD FORONE / F7.0
DCA 57
TAD DECONE
DCA 60
DCA 55 /0 LEFT IN AC,PUT IN 55 SUPPRESS CR/LF
JMS I 7
FGET DAYS
FOUT
FGET HOURS
FOUT
FGET MINS
FOUT
FGET SECS
FOUT
FEXT
CLA CLL
TAD CVS
SZA /-0?
JMP NO1 /NO
IAC /YES
IAC /2 IN AC
JMP VALOUT
NO1, CLA CLL / IS IT 2?
CLA CLL CMA RAL / -2IN AC
TAD CVS
SZA /IS CVS=2?
JMP NO2 /NO
CLA CLL /YES
CLA CLL IAC RTL / PUT 4 IN AC
JMP VALOUT
NO2, CLA CLL
CLA CLL IAC RTL
CIA
TAD CVS
SZA /IS=4?
JMP NO3 /NO =5 STATE
CLA CLL /YES
TAD CVS
IAC
JMP VALOUT
NO3, CLA CLL
JMP VALOUT
VALOUT, DCA 44 /ENTER WITH VALVE STATE IN AC
JMS I 7

```

ORIGINAL PAGE IS  
OF POOR QUALITY

```

FLOT
FOUT  /OUT PUT LEVEL IN OCTAL VALVE CODE
FEXT
CLA CLL
JMP I POUTDT
FORONE, 0007
DECONE,0000
*76
POUTDT, OUTDT
*3000
OUTDT, CLA CLL
TLS
JMS I PTYPX
OPMS2 / 1CR-LF
CLA CLL
TAD FORTWO
DCA 57
TAD DECTWO
DCA 60 / F8.1
TAD LSN
DCA 4
JMS I 7 /# OF SAMPLES PER LEVEL PER INSTR.
FLOT
FPUT FPLSN
FGET INDXR1
FPUT R16
OPLOP1, FGET I R16
FDIV FPLSN
FOUT
FISZ R20 /DONE 1ST HALF?
FJMP OPLOP1 /NO
FEXT /YES
CLA CLL
JMP I PTYPX
OPMS2
CLA CLL
JMS I 7
FGET INDXR2
FPUT R16
OPLOP2, FGET I R16
FDIV FPLSN
FOUT
FISZ R20
FJMP OPLOP2 /NO
FEXT /YES ; LAST HALF DONE
CLA CLL
JMP I PTYPX
OPMS2
JMP I PTYPX
OPMS1
CLA CLL
JMS I 7
FGET INDXR3
FPUT R16
ZERLOP, FGET ZER000

```

FPUT I R16

FISZ R20 /DONE ALL INSTR?

FJMP ZERLOP /NO

FEXT

CLA CLL

JMP I PEXIT

OPMS1, 3737 /2CR-LF

0000 /00

OPMS2, 3700 /CR-LF0

FORTWO, 0010

DECTWO, 0001/ F8.1 FORMAT

FPLSN, 0;0;0

INDXR1, BUFF-3;0;-7

INDXR2, BUFF+22;0;-7

\*16

R16,0

R17,0

R20,0

\*30

INDXR3, BUFF-3; BUFF-3; -16

PAUSE

CSIF=6145

SDTAC=6146

\*600

SCNSER, CLA CLL

CSIF /CLEAR SCANNER INTERRUPT FLAG

CLA CLL

SDTAC

DCA TEMDAT

TAD TEMDAT

AND MASLSD /CK TO SEE IF <0,MASLSD=0001

SZA /SKIP IF AC=0 ONLY

JMP UP

CLA CLL /YES NUM<0

TAD TEMDAT

JMS I PBCDEN

CIA /CONVERT AND NEGATE

DCA TEMDAT

JMP DATAIN

UP, CLA CLL /NO IT IS >0

TAD TEMDAT

JMS I PBCDEN

DCA TEMDAT  
JMP DATAIN

TEMDAT,0  
MASLSD,0001

DATAIN, CLA CLL  
TAD TEMDAT  
DCA 44  
JMS I 7

FLOT

FMPY NCSLOP

/ SLOPE IS ONE TO ONE FOR NUMBER SCALE

FADD NCZERO

/ INTERCEPT IS +998(USED 1000 TEMP)

FADD I R16

FPUT I R17

FISZ R20

/FINISHED ALL, INST?

FJMP NOTYET

FGET INDXR3

/YES RESET INDEX REG.

FPUT R16

FEXT

JMP SNDCHN

NOTYET, FEXT

JMP SNDCHN

NCSLOP, 0001; 2000; 0000

NCZERO, 0012; 3720; 0000

SNDCHN, CLA CLL

ISZ CHAN /IS CHAN -0?

JMP EX1 /NO

TAD NI /YES, SO RESET CHAN

CIA /NEGATE

DCA CHAN

ISZ NSL /IS NSL=0?

JMP EX1 /NO

CLA CLL /YES, RESET NSL

TAD LSN

CIA

DCA NSL

NOP /WAIT FOR LAST CLOCK INTERRUPT

JMP I PEXIT

EX1, CLA CLL

TAD CHAN

CIA

CNTSN / SEND TO SCANNER

JMP I PEXIT

EXIT, CLA CLL

CAF

TAD FLAGS

RTF

CLA

TAD AC

JMP I 0

ERROR, CLA CLL

A-6

ORIGINAL PAGE IS  
OF POOR QUALITY



CAF /CLEAR ALL FLAGS

TLS

JMP I PTYPX

ERRMEI

CLA CLL

CAF

JMP I PEXIT

ERRMEI, 0522 /ER

2217 /RO

2240 /R

1116 /IN

4023 / S

1311 / KI

2040 /P

0310 /CH

0111 /AI

1640 /N

3700 /-0

\*115

PEXIT,EXIT

PERROR, ERROR

PAUSE

SICFS=6137

SISFS=6147

CNTSN=6141

CCIF=6135

RCW1=6131

RCW2=6132

RCW3=6133

\*1000

SERVE, DCA AC

GTF

DCA FLAGS

CLL /SKIP CHAIN

SICFS /SKIP IF CLOCK FLAG SET

SKP

JMP CLKSER

SISFS / SKIP IF SCANNER FLAG SET

SKP

JMP I PSCNSR

KSF /KEYBOARD SET FLAG?

SKP

JMP I PTTYSR

JMP I PERROR

CAF

JMP I PEXIT / SHOULD NEVER GET HERE

CLKKSER, CLA CLL

ORIGINAL PAGE IS  
OF POOR QUALITY

```

    CCIF /CLEAR CLOCK INTERRUPT FLAG
    ISZ CIC /IS CIC-0?
    SKP /NO
    JMP CLKRED /YES
    CLA CLL
    IAC /PUT 1 IN AC
    TAD CIC
    SMA
    JMP NOVOFF
RENTER, CLA CLL
    JMS I 7
    FGET INDXR3
    FPUT R16
    FEXT
    CLA CLL
/ THIS RESETS INDEX FOR ADDING TO THE DATA FILES
    TAD NI
    CIA
    DCA CHAN
    TAD CHAN
    CIA
    CNTSN /START CONVERSION ON 1ST INST
    CLA CLL
    JMP I PEXIT
NOVOFF, CLA CLL
    TAD CVS
    SACTV / SEND TO VALVES
    JMP RENTER
CLKRED, CLA CLL
    CCIF /CLEAR CK INTR. FLOG
    RCW1 /READ 1ST CLK WORD
    DCA FSTWRD
    RCW2 /READ 2ND WORD
    DCA SNDWRD
    RCW3
    DCA THRWRD
    TAD MINS5
    CIA / NEGATE
    DCA CIC / RESET THE CIC COUNTER
    JMP CLKPRO
FSTWRD,0
SNDWRD,0
THRWRD,0
CLKPRO, CLA CLL
    TAD FSTWRD
    JMS MSK1 /SAME PAGE
    DCA UNSSEC
    TAD FSTWRD
    JMS MSK2 /SAME PAGE
    DCA TNSSEC
    TAD FSTWRD
    JMS MSK3 /SAME PAGE

```

DCA UNSMIN

CLL

TAD SNDWRD /2ND WRD PROCESSOR

JMS MSK1

DCA TNSMIN

TAD SNDWRD

JMS MSK2

DCA UNSHRS

TAD SNDWRD

JMS MSK3

DCA TNSHRS

CLL

TAD THRWRD /3RD WORD

JMS MSK1

DCA UNSDYS

TAD THRWRD

JMS MSK2

DCA TNSDYS

TAD THRWRD

JMS MSK3

DCA HNDDYS

CLL

JMP I PDTMSR

MSK1,0

AND MASKT1

CLL

RTR

RTR

RTR

RTR

JMS I PBCDBN

JMP I MSK1

MSK2,0

AND MASKT2

CLL

RTR

RTR

JMS I PBCDBN

JMP I MSK2

MSK3,0

AND MASKT3

CLL

JMS I PBCDBN

JMP I MSK3

MASKT1,7400

MASKT2,0350

MASKT3,0017

\*72

CVS,0

\*70

AC,0

\*101

PBCDBN,BCDBIN

\*160

UNSSSEC,0

TNSSEC,0

UNSMIN,0  
TNSMIN,0  
UNSHRS,0  
TNSHRS,0  
UNSDYS,0  
TNSDYS,0  
HNDDYS,0

\*102

PSCNSR,SCNSER  
PTTYSR,TTYSER  
PDTMSR,DTMSTR  
PVALVR,VALVER

\*1

JMP I PSERVE

\*5

PSERVE, SERVE  
PAUSE

\*1200

DTMSTR, CLA CLL  
TAD UNSSEC  
DCA 44  
JMS I 7  
FLOT  
FPUT SECS  
FEXT  
CLA CLL  
TAD TNSSEC  
DCA 44  
JMS I 7  
FLOT  
FMPY TEN  
FADD SECS  
FPUT SECS  
FEXT  
CLA CLL  
TAD UNSMIN  
DCA 44  
JMS I 7  
FLOT  
FPUT MINS  
FEXT  
CLA CLL  
TAD TNSMIN  
DCA 44  
JMS I 7  
FLOT  
FMPY TEN  
FADD MINS  
FPUT MINS  
FEXT  
CLA CLL  
TAD UNSHRS  
DCA 44

JMS I 7  
 FLOT  
 FPUT HOURS  
 FEXT  
 CLA CLL  
 TAD UNSDYS  
 DCA 44  
 JMS I 7  
 FLOT  
 FMPY TEN  
 FADD HOURS  
 FPUT HOURS  
 FEXT  
 CLA CLL  
 TAD UNSDYS  
 DCA 44  
 JMS I 7  
 FLOT  
 FPUT DAYS  
 FEXT  
 CLA CLL  
 TAD TNSDYS  
 DCA 44  
 JMS I 7  
 FLOT  
 FMPY TEN  
 FADD DAYS  
 FPUT DAYS  
 FEXT  
 CLA CLL  
 JMP I PYEARS

TEN, 4; 2400; 0  
 BCDBIN, 0

DCA TEMPH  
 TAD TEMPH  
 AND LDIGIT  
 CLL RTR  
 DCA CUNT  
 TAD CUNT  
 RAR  
 TAD CUNT  
 CMA IAC  
 TAD TEMPH  
 DCA TEMPH  
 TAD TEMPH  
 AND MDIGIT  
 CLL RTR  
 DCA CUNT  
 TAD CUNT  
 RAR  
 TAD CUNT  
 CMA IAC  
 TAD TEMPH

```

      JMP I BCDBIN
LDIGIT, 7400
MDIGIT, 7760
CUNT, 0
TEMPH, 0
      *76
PYEARS, YEARS
      *3200
YEARS,  CLA CLL
      TAD HNDDYS
      DCA 44
      JMS I 7
      FLOT.
      FMPY HUN100
      FADD DAYS
      FPUT DAYS
      FEXT
      CLA CLL
      JMP I PVALVR
HUN100, 7; 3100; 0
      *123
SECS,0;0;0
MINS,0;0;0
HOURS,0;0;0
DAYS,0;0;0
      PAUSE

```

```

      *120
ZER000, 0 ;0 ;0
      *2400
ZERO,  CLA CLL
      TAD NCLUT
      CIA
      DCA TULCN
      JMP I PNOVAL

```

/ THIS ROUTINE WILL BE WRITTEN LATER(MAY 23,1974)  
 / CURRENT SYSTEM WILL NOT ZERO BUT ONLY A SMALL CHANGE IS  
 / NEEDED.

```

      *112
PNOVAL, NOVAL
      *71
FLAGS, 0
      PAUSE

```

SACTV=6151

\*1400

VALVER, CLA CLL

ISZ VL /IS VL=0?

JMP VLNZ /NO

TAD NLV /YES,NLV=+4=# LEVELS

CIA

DCA VL /RESET VL

CLA CLL

IAC

RTL

IAC

DCA CVS

TAD CVS / CURRENT VALVE STATE(A,B,C)

SACTV /PUT 101 ON AC, SEND TO VALVES

ISZ TULCN /READY TO ZERO?

JMP NOVAL / NO; NCLUT=# CYCLES UP TOWER

JMP I PZERO /YES

VLNZ, CLA CLL /VL=-1.-2,-3

IAC

IAC

TAD VL /PUT +2 IN AC, TAD VL, RES +,0,-

SPA

JMP SEVFIV /VL=-3,75 FOOT

SZA

JMP TWFIV /VL=-1,25 FEET

CLA CLL

IAC

IAC

DCA CVS

TAD CVS

SACTV /SET 010 IN AC, SEND TO VALVES,50 FEET

JMP NOVAL

SEVFIV, CLA CLL

IAC

RTL / 100 IN AC

DCA CVS

TAD CVS

SACTV /75 FEET

JMP NOVAL

TWFIV, CLA CLL

DCA CVS

TAD CVS

SACTV /000 IN AC , SEND TO VALVES, 25 FEET

JMP NOVAL

NOVAL, CLA CLL

IAC /NO VALVE TURN ON

RTL

RAL / MAKE 1000 IN AC

TAD CVS / ADD CURRENT VALVE STATE

SACTV

CLA CLL

ORIGINAL PAGE IS  
OF POOR QUALITY

```

      JMP I PDATOT
ROTATE,   CLA CLL CML
      NOP
BEGIN, DCA SAVEAC
      RAL
      DCA SAVEL
      TAD MASK
      OSR
      DCA COUNT
      OSR
      RAL
      SZL CLA
      JMS LEFT
      JMS RIGHT
      CLL
GO, TAD SAVEL
      RAR
      TAD SAVEAC
INSTR,   RAR
      ISZ COUNTR
      JMP .-1
      ISZ COUNT
      JMP .-3
      JMP BEGIN
SAVEAC,0
SAVEL,0
MASK, 7000
COUNTR,0
COUNT,0
LEFT,0
      ISZ LEFT
      TAD KRAL
      DCA INSTR
      JMP I LEFT
RIGHT,0
      TAD KRAR
      DCA INSTR
      JMP I RIGHT
KRAR, 7010
KRAL, 7004
      *73
NCLUT,0
TULCN, 0
      *106
PZERO,ZERO
      *107
PDATOT, DTAOUT
PAUSE

```

ORIGINAL PAGE IS  
OF POOR QUALITY



```

START,  *1600
        IOF
KCC      /CLEAR KEYBOARD FLAG
CLA CLL
TLS
JMS I PTYPX
MSG01
JMS I PTYPX
MSG02
JMS INPT
CLA CLL
TAD 150
DCA NI
TAD NI
CIA
DCA CHAN / SET CHAN=-NI
CLA CLL
JMS I PTYPX
MSG03
JMS INPT
CLA CLL
TAD 150
DCA NLV / SET UP NLV(#OF LEVELS)
TAD NLV
CIA
DCA VL /SET UP VL=-NLV
CLA CLL
JMS I PTYPX
MSG04
JMS INPT
CLA CLL
TAD 150
DCA MINS5 / SET UP TIME / LEVEL(MINS=5 TYPICAL)
TAD MINS5
CIA
DCA CIC / SET UP CIC
CLA CLL
JMS I PTYPX
MSG05
JMS INPT
CLA CLL
TAD 150
DCA LSN
TAD LSN
CIA
DCA NSL /SET NSL
CLA CLL
JMS I PTYPX
ZERCYM
JMS INPT
CLA CLL
TAD 150

```

DCA NCLUT  
 TAD NCLUT  
 CIA  
 DCA TULCN  
 CLL  
 JMP PRESTR  
 INPT, 0 /THIS PAGE INPUT OF NUMERICAL CONSTANTS  
 CLA CLL  
 JMS I 7  
 FIN  
 FFIX  
 FPUT 150  
 FEXT  
 CLA CLL  
 JMP I INPT  
 MSG01, 3737 /--  
 4040  
 4040  
 1140 /I  
 0115 /AM  
 4015 / M  
 1716 /ON  
 0122 /AR  
 0310 /CH  
 3700 /-0  
 MSG02, 1116 /IN  
 2025 /PU  
 2440 /T  
 2410 /TH  
 0540 /E  
 1625 /NU  
 1502 /MB  
 0522 /ER  
 4017 / O  
 0640 /F  
 0401 /DA  
 2401 /TA  
 4020 / P  
 1722 /OR  
 2423 /TS  
 3700 /-0  
 MSG03, 1116 /IN  
 2025 /PU  
 2440 /T  
 2410 /TH  
 0540 /E  
 1625 /NU  
 1502 /MB  
 0522 /ER  
 4017 / O  
 0640 /F

ORIGINAL PAGE IS  
 OF POOR QUALITY

1405 /LE  
2605 /VE  
1423 /LS  
3700 /-0

PRESTR, CLA CLL

IAC  
RTL  
IAC  
DCA CVS  
TAD CVS  
SACTV /SET UP 15 LEVEL

CLA  
IAC  
RTL  
RAL  
TAD CVS  
SACTV  
CLA CLL  
JMP I PENTER  
\*2700

ZERCYM, 1116 /IN

2025 /PU  
2440 /T  
2410 /TH  
0540 /E  
1625 /NU  
1502 /NB  
0522 /ER  
4017 / O  
0640 /F  
2411 /TI  
1505 /ME  
2340 /S  
2520 /UP  
4024 / T  
1727 /OW  
0524 /ER  
3702 /-B  
0524 /ET  
2705 /WE  
0516 /EN  
4032 / Z  
0522 /ER  
1740 /O  
0331 /CY  
0314 /CL  
0523 /ES  
3700 /-0

PAUSE

\*2000

TYPX, 0  
CLA CLL  
TAD I TYPX  
DCA TYPNT  
ISZ TYPX  
TYPX1, TAD I TYPNT  
RTR  
RTR  
RTR  
JMS TYPY  
TAD I TYPNT  
ISZ TYPNT  
JMS TYPY  
JMP TYPX1

TYPNT, 0  
TYPY, 0  
AND TK77  
SNA  
JMP I TYPX  
TAD TKM37  
SZA  
JMP TYPY1  
TAD TK215  
JMS TLSX  
TAD TKM125

TYPY1, SPA  
TAD TK100  
TAD TK237  
JMS TLSX  
JMP I TYPY

TK77, 77  
TKM37, -37  
TK215, 215  
TKM125, -125  
TK100, 100  
TK237, 237  
TLSX, 0

TSF  
JMP --1  
TLS  
CLA  
JMP I TLSX

KRBX, 0  
KSF  
JMP --1  
KRB  
JMP I KRBX

KREAD, 0  
CLA CLL  
TAD I KREAD  
ISZ KREAD  
DCA KRPNT  
TAD I KREAD  
DCA KRCNT  
KRB1, JMS KRBX  
DCA I KRPNT  
TAD KRTAB

KRB3, DCA KRBKS  
 TAD I KRPNT  
 ISZ KRBKS  
 SNA CLA  
 JMP I KRBKS  
 ISZ KRBKS  
 TAD I KRBKS  
 SZA  
 JMP KRB3  
 JMS KRBKS  
 ISZ KRCNT  
 JMP KRB6  
 TAD TK207  
 KRB5, JMS TLSX  
 CLA CMA  
 TAD KRCNT  
 JMP KRB1-1  
 KRB6, TAD I KRPNT  
 ISZ KRPNT  
 JMS TLSX  
 JMP KRB1  
 KRUB, CLA CMA  
 JMS KRBKS  
 ISZ KRFLAG  
 TAD I KREAD  
 CIA  
 TAD KRCNT  
 SNA CLA  
 JMP KRUB1  
 CLA CMA  
 TAD KRPNT  
 DCA KRPNT  
 TAD I KRPNT  
 JMP KRB5  
 KRUB1, TAD TK237  
 JMS TYPY  
 JMS KRBKS  
 JMP KRB1  
 KRCR, JMS KRBKS  
 TAD TK237  
 JMS TYPY  
 DCA I KRPNT  
 ISZ KREAD  
 JMP I KREAD  
 KRBKS, 0  
 TAD KRFLAG  
 SZA CLA  
 TAD TK334  
 SZA  
 JMS TLSX

DCA KRFLAG  
JMP I KRBKS

KRFLAG, 0  
KRPNT=TYPNT  
KRCNT=TYPX  
TK207, 207  
TK334, 334  
KRTAB, .

JMP KRB1  
-200;  
JMP KRB1  
-212;  
JMP KRB1  
-215;  
JMP KRCR  
-377;  
JMP KRUB  
0

PAUSE

\*2600

MESG04, 1116 /IN  
2025 /PU  
2440 /T  
2410 /TH  
0540 /E  
2411 /TI  
1505 /ME  
5015 /CM  
1116 /IN  
5140 /)  
2320 /SP  
0516 /EN  
2440 /T  
2005 /PE  
2240 /R  
1405 /LE  
2605 /VE  
1440 /L  
3700 /-@

MESG05, 1116 /IN  
2025 /PU  
2440 /T  
1625 /NU  
1502 /MB  
0522 /ER  
4017 / O  
0640 /F  
2301 /SA  
1520 /MP  
1405 /LE  
2340 /S  
2401 /TA  
1305 /KE

1637 /N-  
0231 /BY  
4005 / E  
0103 /AC  
1040 /H  
1116 /IN  
2324 /ST  
2225 /RU  
1505 /ME  
1624 /NT  
4020 / P  
0522 /ER  
4014 / L  
0526 /EV  
0514 /EL  
3700 /-0

\*100  
PTYPX, TYPX  
\*140

NI,0  
CHAN,0  
NLV,0  
VL,0  
MINS5,0  
CIC,0  
LSN,0  
NSL,0  
PAUSE

FIXMRI FJMP=0000  
FIXMRI FJMS=7000  
FISZ=0000  
FEXT=0000  
FSQU=0001  
FSQR=0002  
FSIN=0003  
FCOS=0004  
FATN=0005  
FEXP=0006  
FLOG=0007  
FNEG=0010  
FIN =0011  
FOUT=0012  
FFIX=0013  
FLOT=0014  
FNOR=7000  
FCDF=7001  
FSW0=7002  
FSW1=7003  
FHLT=7004  
FSMA=7110  
FSZA=7050  
FSPA=7100  
FSNA=7040  
FNOP=7010  
FSKP=7020  
PAUSE